S. Pauli *Res Math Sci* (2023) 10:26
https://doi.org/10.1007/s40687-023-00392-0

Research in the Mathematical Sciences

RESEARCH

Computing $\mathbb{A}^1$-Euler numbers with Macaulay2

Sabrina Pauli*

*Correspondence: sabrina.pauli@hhu.de
Mathematisches Institut der Heinrich-Heine-Universität Düsseldorf, Düsseldorf, Germany

Abstract

We use Macaulay2 for several enriched counts in GW(k). First, we compute the count of lines on a general cubic surface using Macaulay2 over $\mathbb{F}_p$ in GW($\mathbb{F}_p$) for p a prime number and over $\mathbb{Q}$ in GW($\mathbb{Q}$). This gives a new proof for the fact that the $\mathbb{A}^1$-Euler number of $\mathrm{Sym}^3\,\mathcal{S}^* \to \mathrm{Gr}(2,4)$ is $15\langle 1\rangle + 12\langle -1\rangle$. Then, we compute the count of lines in $\mathbb{P}^3$ meeting 4 general lines, the count of lines on a quadratic surface meeting one general line and the count of singular elements in a pencil of degree d-surfaces. Finally, we provide code to compute the EKL-form and compute several $\mathbb{A}^1$-Milnor numbers.

1 Introduction

In [8] Kass and Wickelgren count the lines on a smooth cubic surface as an element of the Grothendieck-Witt ring $\mathrm{GW}(k)$ of a field k by computing the $\mathbb{A}^1$-Euler number of the vector bundle $\mathcal{E} := \mathrm{Sym}^3\,\mathcal{S}^* \to \mathrm{Gr}(2,4)$ which is by definition the sum of the local indices, that is the local $\mathbb{A}^1$-degrees, at the zeros of a general section. Here, $\mathrm{Gr}(2,4)$ denotes the Grassmannian of lines in $\mathbb{P}^3$ and $\mathcal{S} \to \mathrm{Gr}(2,4)$ its tautological bundle.

For a field L, denote by $\mathcal{E}_L$ the base change of $\mathcal{E}$ to L. Let $F \in \mathbb{F}_p[X_0, X_1, X_2, X_3]_3$ be a random homogeneous degree 3 polynomial in 4 variables. Then F defines a general cubic surface $X = \{F = 0\} \subset \mathbb{P}^3_{\mathbb{F}_p}$ and a section σ_F of $\mathcal{E}_{\mathbb{F}_p}$ by restriction. The zeros of σ_F are the lines on X.

Let $\mathbb{A}^4_{\mathbb{F}_p} = \mathrm{Spec}(\mathbb{F}_p[x_1, x_2, x_3, x_4]) \subset \mathrm{Gr}(2,4)$ be the open affine subset of the Grassmannian consisting of the lines spanned by $x_1e_1 + x_3e_2 + e_3$ and $x_2e_1 + x_4e_2 + e_4$ where (e_1, e_2, e_3, e_4) is the standard basis for $\mathbb{F}_p^4$. For the general cubic surface X, all lines on X are elements of this open affine subset of $\mathrm{Gr}(2,4)$ and hence the $\mathbb{A}^1$-Euler number $e^{\mathbb{A}^1}(\mathcal{E}_{\mathbb{F}_p}) \in \mathrm{GW}(\mathbb{F}_p)$ (or the count of lines on the cubic surface X) can be computed as the sum of local $\mathbb{A}^1$-degrees of the zeros of $\sigma_f|_{\mathbb{A}^4} = (f_1, f_2, f_3, f_4) : \mathbb{A}^4 \to \mathbb{A}^4$ by [9].

The $\mathbb{F}_p$-algebra $\frac{\mathbb{F}_p[x_1,x_2,x_3,x_4]}{(f_1,f_2,f_3,f_4)}$ is 0 dimensional and thus there are finitely many lines on X. Call these lines $l_1, \ldots, l_n$. By [8, Corollary 51] the lines on a general and thus smooth cubic surface are simple. This means that the lines $l_1, \ldots, l_n$ are simple zeros of $(f_1, f_2, f_3, f_4) : \mathbb{A}^4_{\mathbb{F}_p} \to \mathbb{A}^4_{\mathbb{F}_p}$. It follows that $\mathbb{F}_p[x_1, x_2, x_3, x_4]/I$ is isomorphic to the product of fields $L_1 \times \cdots \times L_n$ where $L_j = \mathbb{F}_p[x_1, x_2, x_3, x_4]/\mathfrak{m}_j$ is the field of definition of l_j (that

Springer

is residue field of the point in $\mathrm{Gr}(2,4)$ corresponding to l_j) for $j=1,\dots,n$. By [9, Lemma 9] the local index of l_j is equal $\langle J_{L_j}\rangle \in \mathrm{GW}(L_j)$ where J_{L_j} is the image of the jacobian element $J := \det \frac{\partial f_i}{\partial x_l}$ in $L_j = \mathbb{F}_p[x_1,x_2,x_3,x_4]/\mathfrak{m}_j$ and it follows that the $\mathbb{A}^1$-Euler number of $\mathrm{Sym}^3\mathcal{S}^* \to \mathrm{Gr}(2,4)$ is given by

$$e^{\mathbb{A}^1}(\mathcal{E}_{\mathbb{F}_p}) = \sum_{j=1}^{n} \mathrm{Tr}_{L_j/\mathbb{F}_p}(\langle J_{L_j}\rangle) \in \mathrm{GW}(\mathbb{F}_p). \tag{1}$$

We use Macaulay2 to compute the rank and discriminant of (1) when $p = 32003$. The computation gives an element in $\mathrm{GW}(\mathbb{F}_{32003})$ of rank 27 and discriminant $1 \in \mathbb{F}^*_{32003}/(\mathbb{F}^*_{32003})^2$. Two elements in $\mathrm{GW}(\mathbb{F}_{32003})$ are equal if and only if they have the same rank and discriminant, so this determines the count of lines on a cubic surface in $\mathrm{GW}(\mathbb{F}_{32003})$ completely.

Similarly, we use Macaulay2 to get the Gram matrix of the form $e^{\mathbb{A}^1}(\mathcal{E}_{\mathbb{Q}}) \in \mathrm{GW}(\mathbb{Q})$ over the rational numbers $\mathbb{Q}$. We view $e^{\mathbb{A}^1}(\mathcal{E}_{\mathbb{Q}})$ as a bilinear form over the real numbers $\mathbb{R}$ and compute its signature which is equal to 3.

By Theorem 5.8 in [1] $e^{\mathbb{A}^1}(\mathcal{E}) = e^{\mathbb{A}^1}(\mathrm{Sym}^3\mathcal{S}^*)$ is equal to either

$$\frac{n_{\mathbb{C}}+n_{\mathbb{R}}}{2}\langle 1\rangle + \frac{n_{\mathbb{C}}-n_{\mathbb{R}}}{2}\langle -1\rangle \in \mathrm{GW}(k) \tag{2}$$

or

$$\frac{n_{\mathbb{C}}+n_{\mathbb{R}}}{2}\langle 1\rangle + \frac{n_{\mathbb{C}}-n_{\mathbb{R}}}{2}\langle -1\rangle + \langle 2\rangle - \langle 1\rangle \in \mathrm{GW}(k) \tag{3}$$

for $n_{\mathbb{C}}, n_{\mathbb{R}} \in \mathbb{Z}$ and a field k. By [1, Remark 5.7] $n_{\mathbb{C}}$ and $n_{\mathbb{R}}$ are the Euler numbers of the real and complex bundle, respectively. The complex count $n_{\mathbb{C}}$ is equal to the rank of our form which is $n_{\mathbb{C}} = 27$, and the real count is equal to the signature, so $n_{\mathbb{R}} = 3$. In [12, ♧8] and [1, Corollary] it is shown that the $\mathbb{A}^1$-Euler number of direct sums of symmetric power of the dual tautological bundle on a Grassmannian is always of form (2) when defined, using the theory of Witt-valued characteristic classes. The proof here is independent of this theory and we may also apply it to bundles which are not of this form.

Since 2 is not a square for our chosen prime 32003, we can rule out (3) for the count of lines on a cubic surface and hence we have a new proof of the fact that

$$e^{\mathbb{A}^1}(\mathrm{Sym}^3\mathcal{S}^*) = 15\langle 1\rangle + 12\langle -1\rangle \in \mathrm{GW}(k) \tag{4}$$

which is the main result in [8]. The complex count $n_{\mathbb{C}}$ is the classical result by Cayley and Salmon that there are 27 lines on a smooth cubic surface [3]. Segre studied the real lines on a smooth cubic surface in [16]. See also [7,14] for the real count.

Similarly, we get an enriched count of lines meeting 4 general lines in $\mathbb{P}^3$ (this has already been computed in [17]) and of lines on a quadratic surface meeting one general line by computing the $\mathbb{A}^1$-Euler numbers $e^{\mathbb{A}^1}(\bigoplus_{i=1}^4 \wedge^2\mathcal{S}^* \to \mathrm{Gr}(2,4))$ and $e^{\mathbb{A}^1}(\wedge^2\mathcal{S}^* \oplus \mathrm{Sym}^2\mathcal{S}^* \to \mathrm{Gr}(2,4))$, respectively. Note, that neither of these vector bundles is a direct sum of symmetric powers of the dual tautological bundle and we cannot use [12, ♧8] and [1, Corollary] to rule out (3). However, we already know that the $\mathbb{A}^1$-Euler number of both of these bundles will be a multiple of the hyperbolic form $\mathbb{H} = \langle 1\rangle + \langle -1\rangle$ since they have direct summands of odd rank [17, Proposition 12].

Furthermore, we count singular elements on a pencil of degree d surfaces as the $\mathbb{A}^1$-Euler number of $\oplus_{i=1}^4 \pi_1^*\mathcal{O}_{\mathbb{P}^3}(d-1) \otimes \pi_2^*\mathcal{O}_{\mathbb{P}^1}(1) \to \mathbb{P}^3 \times \mathbb{P}^1$.

Finally, we provide code for computing the EKL-form (see [9]) which computes the local $\mathbb{A}^1$-degree for non-simple zeros.

In the appendix we compute the $\mathbb{A}^1$-Milnor numbers of several Fuchsian singularities and provide one explicit example of the Gram matrix of a form representing $e^{\mathbb{A}^1}(\mathcal{E}_{\mathbb{F}_{11}}) \in \mathrm{GW}(\mathbb{F}_{11})$.

2 $\mathbb{A}^1$-Euler numbers

2.1 Definition of the $\mathbb{A}^1$-Euler number

Let k be a field and let $\pi : E \to X$ be a vector bundle of rank r over a smooth and proper scheme X of dimension r. Assume further that for each closed point $x \in X$ there is a Zariski neiborhood U of x which is isomorphic to affine space $\mathbb{A}^r$.

Remark 1 In our examples, X is either a Grassmannian of lines or projective space which both have standard coverings by open affine subsets $U \cong \mathbb{A}^r$. All definitions also work when X does not admit a Zariski covering by affine spaces. Then one needs *Nisnevich coordinates* [8, Definition 17 and Lemma 18].

We recall the definition of the $\mathbb{A}^1$-Euler number of $\pi : E \to X$ from [8, §4]. Recall that a *(weak) orientation* of E is an isomorphism $\phi : \det E \cong L^{\otimes 2}$ where $L \to X$ is a line bundle.

Definition 1 A *relative orientation* of E is a orientation of the line bundle $\mathrm{Hom}(\det TX, \det E)$, that is, an isomorphism $\phi : \mathrm{Hom}(\det TX, \det E) \xrightarrow{\cong} L^{\otimes 2}$ where $TX \to X$ denotes the tangent bundle of X and $L \to X$ is a line bundle.

Remark 2 If both the tangent bundle of X and E are orientable, then E is relatively orientable since $\mathrm{Hom}(\det TX, \det E) \cong (\det TX)^{-1} \otimes \det E$. However, $\pi : E \to X$ can still be relatively orientable even though E and TX are not.

Assume that $\pi : E \to X$ is equipped with a relative orientation ϕ. An open affine subset $\psi : U \cong \mathbb{A}^r$ of X defines a trivialization of $TX|_U$.

Definition 2 A trivialization of $E|_U$ with $\psi : U \cong \mathbb{A}^r$ is *compatible* with the relative orientation ϕ and ψ if the element of $\mathrm{Hom}(\det TX|_U, \det E|_U)$ sending the distinguished basis element of $\det TX|_U$ to the distinguished element of $\det E|_U$ is sent to a square by ϕ.

Let $\sigma : X \to E$ be a section of E with an isolated zero $x \in X$. We now define the local index $\mathrm{ind}_x \sigma$ of σ at x, that is the local contribution of the zero x to the $\mathbb{A}^1$-Euler number. Choose a neighborhood $x \in U$ of x which is isomorphic to affine space $\psi : U \cong \mathbb{A}^r$ and a trivialization $E|_U \cong \mathbb{A}^r \times \mathbb{A}^r$ compatible with the chosen relative orientation ϕ. Locally the following composition

$$U \overset{\psi}{\cong} \mathbb{A}^r \xrightarrow{\sigma|_U} E|_U \cong \mathbb{A}^r \times \mathbb{A}^r \xrightarrow{\pi_2} \mathbb{A}^r$$

where the second map is the projection onto the second factor, is given by r regular functions $(f_1, \ldots, f_r) : \mathbb{A}^r \to \mathbb{A}^r$.

The *local index* $\mathrm{ind}_x \sigma$ of σ at x is the local $\mathbb{A}^1$-degree $\deg_x^{\mathbb{A}^1}(f_1, \ldots, f_r)$ of $(f_1, \ldots, f_r) : \mathbb{A}^r \to \mathbb{A}^r$ at x. For the definition of the local $\mathbb{A}^1$-degree we refer to [9, §2].

We define the $\mathbb{A}^1$-*Euler number* $e^{\mathbb{A}^1}(E, \sigma)$ with respect to a section $\sigma : X \to E$ with only isolated zeros to be sum of indices of the zeros of σ. It turns out that $e^{\mathbb{A}^1}(E, \sigma)$ does not depend on the chosen section [1, Theorem 1.1] and we can define the $\mathbb{A}^1$-Euler number independently of σ.

Definition 3 Let $\pi : E \to X$ be a vector bundle of rank r equal to the dimension of the smooth, proper scheme X over a field k equipped with a relative orientation, then the $\mathbb{A}^1$*-Euler number* is defined by $e^{\mathbb{A}^1}(E) := e^{\mathbb{A}^1}(E, \sigma)$ for a section σ with only isolated zeros.

2.1.1 Computation of the local indices

Next we recall from [9] how the local $\mathbb{A}^1$-degree can be computed. This also yields a formula for the local indices. Let L/k be a finite separable field extension and let $\beta : V \times V \to L$ be a non-degenerate symmetric bilinear form over L. Then the *trace form* $\mathrm{Tr}_{L/k}(\beta)$ is the form

$$V \times V \xrightarrow{\beta} L \xrightarrow{\mathrm{Tr}_{L/k}} k \tag{5}$$

where $\mathrm{Tr}_{L/k}$ denotes the field trace. Assume $x \in X$ is simple zero, that is the *jacobian element* $\frac{\partial f_i}{\partial x_j}(x)$ at x is non-zero. If x is a rational point, its local degree is equal to $\langle J(x)\rangle \in \mathrm{GW}(k)$. When x is not rational, its local $\mathbb{A}^1$-degree can be computed as the trace form $\mathrm{Tr}_{k(x)/k}(\langle J(x)\rangle) \in \mathrm{GW}(k)$ of $\langle J(x)\rangle \in \mathrm{GW}(k(x))$ for finite separable field extensions $k(x)/k$ by [2].

Remark 3 When $x \in X$ is a non-simple zero, its local $\mathbb{A}^1$-degree can be computed with the *EKL-form* (see Sect. 3).

2.2 Cubic surfaces

We compute the rank and discriminant of the $\mathbb{A}^1$-Euler number of $\mathcal{E} = \mathrm{Sym}^3 \mathcal{S}^* \to \mathrm{Gr}(2, 4)$ over $\mathbb{F}_{32003}$.

```
i1 : P = 32003;
i2 : FF = ZZ/P;
```

We generate a random homogeneous degree 3 polynomial F in 4 variables X_0, X_1, X_2 and X_3.

```
i3 : R = FF[X0,X1,X2,X3];
i4 : F = random(3,R);
```

We replace X_0, X_1, X_2 and X_3 by $x_1 s + x_2$, $x_3 s + x_4$, s and 1, respectively, and define I to be the ideal in $C = \mathbb{F}_{32003}[x_1, x_2, x_3, x_4]$ generated by the coefficients s^3, s^2, s and 1 of $F(x_1 s + x_2, x_3 s + x_4, s, 1)$. That means, we let $\mathrm{Spec}\, C = \mathrm{Spec}(\mathbb{F}_{32003}[x_1, x_2, x_3, x_4]) \subset \mathrm{Gr}(2, 4)$ be the open affine subset consiting of the lines spanned by $x_1e_1 + x_3e_2 + e_3$ and $x_2e_1 + x_4e_2 + e_4$ for the standard basis (e_1, e_2, e_3, e_4) of $\mathbb{F}^4_{32003}$ and we let I be the ideal generated by f_1, f_2, f_3, f_4 where (f_1, f_2, f_3, f_4) is equal to

$$(f_1, f_2, f_3, f_4) : \mathbb{A}^4 \xrightarrow{\sigma|_{\mathbb{A}^4} = (\mathrm{id}, (f_1, f_2, f_3, f_4))} \mathbb{A}^4 \times \mathbb{A}^4 \xrightarrow{\pi_2} \mathbb{A}^4,$$

the restriction of the section σ_F of $\mathcal{E}$ defined by F to the chosen open affine set $\mathrm{Spec}(\mathbb{F}_{32003}[x_1, x_2, x_3, x_4])$.

Remark 4 By [8, Corollary 45] the vector bundle $\mathcal{E}$ is relatively orientable and the open affine subset $\mathrm{Spec}\, C \subset \mathrm{Gr}(2, 4)$ is compatible with this relative orientation.

```
i5 : C = FF[x1,x2,x3,x4];
i6 : S = C[s];
```

```
i7 : g = {x1*s+x2,x3*s+x4,s,1};
i8 : m = map(S,R,g);
i9 : I = sub(ideal flatten entries last coefficients m F, C);
```

We use Macaulay2 to compute the dimension and degree of $C/I = \mathbb{F}_{32003}[x_1, x_2, x_3, x_4]/I$.

```
i10 : dim I

o10 = 0

i11 : degree I

o11 = 27
```

Since there are in general finitely many lines on a cubic surface, the expected dimension of C/I is 0. The degree is the dimension of C/I as a $\mathbb{F}_{32003}$-vector space, that is the rank of the non-degenerate symmetric bilinear form (1) which turns out to be 27 as expected.

Since $Q = C/I$ is zero-dimensional Noetherian and hence Artinian, it is isomorphic to its product of localizations at its maximal ideals

$$Q \cong Q_{\mathfrak{m}_1} \times \cdots \times Q_{\mathfrak{m}_n}.$$

By [8, Corollary 53] $(f_1, f_2, f_3, f_4) : \mathbb{A}^4 \to \mathbb{A}^4$ only has simple zeros, that means that $Q_{\mathfrak{m}_i}$ is a finite field extensions of $\mathbb{F}_{32003}$ equal to the residue fields of the $\mathfrak{m}_i$ for $i = 1, \ldots, n$.

The maximal ideals $\mathfrak{m}_i$ correspond to the finitely many lines $l_1, \ldots, l_n$ on $\{F = 0\} \subset \mathbb{P}^3$. This implies that C/I is isomorphic to the product of fields

$$\mathbb{F}_{32003}[x_1, x_2, x_3, x_3]/\mathfrak{m}_1 \times \cdots \times \mathbb{F}_{32003}[x_1, x_2, x_3, x_4]/\mathfrak{m}_n = L_1 \times \cdots \times L_n \tag{6}$$

where $\mathfrak{m}_i$ is maximal ideal defining l_i as point in Gr(2, 4) and L_i is the field of definition of l_i, i.e., the residue field of l_i in Gr(2, 4), for $i = 1, \ldots, n$.

Remark 5 When we pass to the algebraic closure of $\mathbb{F}_{32003}$ we know that Spec(C/I) has 27 closed points. However, in (6) the number of lines n is not necessarily equal to 27 since in general not all lines will be defined over $\mathbb{F}_{32003}$.

We use a primary decomposition of I to find the $\mathfrak{m}_i$.

```
i12 : L = primaryDecomposition I;
i13 : n = length L;
```

Remark 6 Since the ideals $\mathfrak{m}_i$ are actually primes, the primary ideals in the primary decomposition are the minimal primes and in particular unique, and we can let Macaulay2 compute the minimal primes instead of the the primary decomposition of I. This is much more time efficient. However, if the one of the zeros were not simple, one would need the primary decomposition and then apply the EKL-form (see Sect. 3).

The contribution of the line l_i to (1) is $\mathrm{Tr}_{L_i/\mathbb{F}_32003}(\langle J_{L_i}\rangle)$ where J_{L_i} is the image of the jacobian element $J = \det \frac{\partial f_m}{\partial x_j}$ of I in $L_i = C/\mathfrak{m}_i$. The discriminant of (1) is the product of the discriminants of the forms $\mathrm{Tr}_{L_i/\mathbb{F}_{32003}}(\langle J_{L_i}\rangle)$. By [8, Lemma 58] the discriminant of $\mathrm{Tr}_{L_i/\mathbb{F}_{32003}}(\langle J_{L_i}\rangle)$ is a square in $\mathbb{F}_{32003}$ if J_{L_i} is a square in $L_i = \mathbb{F}_{32003}[x_1, x_2, x_3, x_4]/\mathfrak{m}_i$ when the degree $[L_i : \mathbb{F}_{32003}]$ is odd and if J_{L_i} is a non-square in $L_i = \mathbb{F}_{32003}[x_1, x_2, x_3, x_4]/\mathfrak{m}_i$

when $[L_i : \mathbb{F}_{32003}]$ is even. Since the units $\mathbb{F}_q^*$ of a finite field $\mathbb{F}_q$ with q form the cyclic group of order $q-1$, $\mathbb{F}_q^*/(\mathbb{F}_q^*)^2$ is isomorphic to $\mathbb{Z}/2\mathbb{Z}$. By Fermat's little theorem $b^{q-1} \equiv 1 \mod q$ for $b \in \mathbb{F}_q^*$ and b is a square if and only if $b^{\frac{q-1}{2}} \equiv 1 \mod q$. So to find the discriminant of (1) we compute the product

$$\operatorname{disc}((1)) = \prod_{i=1}^{n} \epsilon_i J_{L_i}^{\frac{p^{[L_i:\mathbb{F}_{32003}]}-1}{2}}$$

where $\epsilon = -1$ when $[L_i : \mathbb{F}_{32003}]$ is even and $\epsilon = 1$ when $[L_i : \mathbb{F}_{32003}]$ is odd.

```
i14 : J = determinant jacobian I;
i15 : disc = 1_FF;
i16 : i=0;
i17 : while i<n do
(if even degree L_i
then
disc=disc*lift(J_(C/L_i)^((P^(degree L_i)-1)//2),FF)*(-1)_FF
else
disc=disc*lift(J_(C/L_i)^((P^(degree L_i)-1)//2),FF); i=i+1);
```

The discriminant of (1) is a square.

```
i18 : disc

o18 = 1
```

2.3 The trace form

The trace form (5) can also be defined when L is a finite étale k-algebra like $C/I = \frac{\mathbb{F}_{32003}[x_1,x_2,x_3,x_4]}{(f_1,f_2,f_3,f_4)}$. In particular, the trace form $\operatorname{Tr}_{(C/I)/\mathbb{F}_{32003}}(\langle J_{C/I}\rangle)$ is a bilinear form over $\mathbb{F}_{32003}$ representing $e^{\mathbb{A}^1}(\mathcal{E}_{\mathbb{F}_{32003}}) \in \mathrm{GW}(\mathbb{F}_{32003})$ where $J_{C/I}$ is the image of the jacobian element in C/I.

The following code computes the trace form $\operatorname{Tr}_{L/k}(\langle J\rangle)$ for *FF* a field and I an ideal in polynomial ring C over *FF* such that C/I is a finite étale algebra over *FF*.

```
i19: traceForm = (C,I,J,FF) -> (
B:=basis(C/I);
r:=degree I;
Q:=(J_(C/I))*(transpose B)*B;
toVector := q -> last coefficients(q,Monomials=>B);
fieldTrace := q -> (M:=toVector(q*B_(0,0));i=1;while i<r do
(M=M|(toVector (q*B_(0,i))) ; i=i+1); trace M);
matrix applyTable(entries Q, q->lift(fieldTrace q,FF)))
```

2.3.1 Lines meeting four general lines in $\mathbb{P}^3$

As an example we compute the count of lines meeting 4 general lines in $\mathbb{P}^3$, i.e., we compute the $\mathbb{A}^1$-Euler number of the bundle $\mathcal{E}_2 := \wedge^2\mathcal{S}^* \oplus \wedge^2\mathcal{S}^* \oplus \wedge^2\mathcal{S}^* \oplus \wedge^2\mathcal{S}^* \to \mathrm{Gr}(2,4)$. We know from [17] that this equal to the *hyperbolic form* $\mathbb{H} := \langle 1\rangle + \langle -1\rangle$.

Clearly, $\det(\wedge^2\mathcal{S}^* \oplus \wedge^2\mathcal{S}^* \oplus \wedge^2\mathcal{S}^* \oplus \wedge^2\mathcal{S}^*) \cong (\wedge^2\mathcal{S}^*)^{\otimes 4}$ and thus the vector bundle $\mathcal{E}_2$ is orientable. The Grassmannian $\mathrm{Gr}(2,4)$ is orientable as well (i.e., its tangent bundle

$T\,\mathrm{Gr}(2,5) \cong \mathcal{S}^* \otimes \mathcal{Q}$ is orientable). Those two orientations yield a relative orientation ϕ : $\mathrm{Hom}(T\,\mathrm{Gr}(2,4), \mathcal{E}_2) \cong L^{\otimes 2}$ of $\mathcal{E}_2$. Over the open affine subset $\mathrm{Spec}(\mathbb{F}_{32003}[x_1, x_2, x_3, x_4]) \subset \mathrm{Gr}(2,4)$ from subsection 2.2 the dual tautological bundle $\mathcal{S}^* \to \mathrm{Gr}(2,4)$ has basis the two monomial s and 1 (where again s is the variable on the line). This basis induces a trivialization of the restriction of $\mathcal{E}_2$ to $\mathrm{Spec}\,\mathbb{F}_{32003}[x_1, \ldots, x_4]$. By [17, Lemma 4] the relative orientation ϕ is compatible with this trivialization over $\mathrm{Spec}\,\mathbb{F}_{32003}[x_1, \ldots, x_4]$.

Let $l_1, \ldots, l_4$ be 4 general lines in $\mathbb{P}^3$ and let a_i, b_i be two independent linear forms cutting out l_i for $i = 1, \ldots, 4$.

```
i20 : a1 = random(1,R);
i21 : b1 = random(1,R);
i22 : a2 = random(1,R);
i23 : b2 = random(1,R);
i24 : a3 = random(1,R);
i25 : b3 = random(1,R);
i26 : a4 = random(1,R);
i27 : b4 = random(1,R);
```

The linear forms a_i and b_i define a section $s_i := a_i \wedge b_i$ of $\wedge^2 \mathcal{S}^*$. A line l in $\mathbb{P}^3$ meets the line l_i if and only if $s_i(l) = 0$ by [17, Lemma5].

```
i28 : s1 = lift((last coefficients m a1)_(0,0)*(last coefficients m b1)_(1,0)
-(last coefficients m a1)_(1,0)*(last coefficients m b1)_(0,0),C);
i29 : s2 = lift((last coefficients m a2)_(0,0)*(last coefficients m b2)_(1,0)
-(last coefficients m a2)_(1,0)*(last coefficients m b2)_(0,0),C);
i30 : s3 = lift((last coefficients m a3)_(0,0)*(last coefficients m b3)_(1,0)
-(last coefficients m a3)_(1,0)*(last coefficients m b3)_(0,0),C);
i31 : s4 = lift((last coefficients m a4)_(0,0)*(last coefficients m b4)_(1,0)
-(last coefficients m a4)_(1,0)*(last coefficients m b4)_(0,0),C);
i32 : I2 = ideal(s1,s2,s3,s4);
i33 : J2 = determinant jacobian I2;
i34 : traceForm(C,I2,J2,FF)
```

Let I_2 be the ideal generated by the sections $s_1, \ldots, s_4$ and $J_2 := \det \frac{\partial s_i}{\partial x_j}$. We compute the trace form $\mathrm{Tr}_{(C/I_2)/FF}(\langle J_2 \rangle)$ (where C is still $\mathbb{F}_{32003}[x_1, x_2, x_3, x_4]$) and get a form of rank 2 and discriminant $-1 \in \mathbb{F}^*_{32003}/(\mathbb{F}^*_{32003})^2$ as expected.

2.3.2 Lines on a degree 2 hypersurface in $\mathbb{P}^3$ meeting 1 general line

We compute the count of lines on a quadratic surface meeting a general line as the $\mathbb{A}^1$-Euler number of $\mathcal{E}_2 := \mathrm{Sym}^2 \mathcal{S}^* \oplus \wedge^2 \mathcal{S}^* \to \mathrm{Gr}(2,4)$.

We have $\det(\mathrm{Sym}^2 \mathcal{S}^* \oplus \wedge^2 \mathcal{S}^*) \cong p^* \mathcal{O}_{\mathbb{P}^5}(4)$ where $p : \mathrm{Gr}(2,4) \hookrightarrow \mathbb{P}^5$ is the Plücker embedding. So $\mathcal{E}_3$ is orientable. Since $\mathrm{Gr}(2,4)$ is orientable, too, we get a relative orientation on $\mathcal{E}_3$. Over the open affine subset $\mathbb{F}_{32003}[x_1, \ldots, x_4] \subset \mathrm{Gr}(2,4)$ from 2.2 we get a trivialization of $\mathcal{E}_3$ coming from trivialization of the dual tautological bundle $\mathcal{S}^* \to \mathrm{Gr}(2,4)$.

Lemma 1 *The trivialization of $\mathcal{E}_3|_{\mathbb{F}_{32003}[x_1,\ldots,x_4]}$ is compatible with the relative orientation of $\mathcal{E}_3$ described above.*

Proof As in [8, Definition 39] we define a basis $\tilde{e}_1 = e_1$, $\tilde{e}_2 = e_2$, $\tilde{e}_3 = x_1 e_1 + x_3 e_2 + e_3$ and $\tilde{e}_4 = x_2 e_1 + x_4 e_2 + e_4$ of $\mathbb{F}_{32003}[x_1, x_2, x_3, x_4]^4$, and let $\tilde{\phi}_1$, $\tilde{\phi}_2$, $\tilde{\phi}_3$ and $\tilde{\phi}_4$ be its dual basis. Here e_1, e_2, e_3, e_4 is a basis of $\mathbb{F}^4_{32003}$. Then the open affine subset of lines spanned by $x_1e_1+x_3e_2+e_3$ and $x_2e_1+x_4e_2+e_4$, $U=\operatorname{Spec}\mathbb{F}_{32003}[x_1,x_2,x_3,x_4]\subset \mathrm{Gr}(2,4)$, yields a basis

$$\tilde{\phi}_3\otimes\tilde{e}_1,\tilde{\phi}_4\otimes\tilde{e}_1,\tilde{\phi}_3\otimes\tilde{e}_2,\tilde{\phi}_4\otimes\tilde{e}_2$$

of $T\,\mathrm{Gr}(2,4)|_U$ and a basis

$$(\tilde{\phi}_3^2,0),(\tilde{\phi}_3\tilde{\phi}_4,0),(\tilde{\phi}_4^2,0),(0,\tilde{\phi}_3\wedge\tilde{\phi}_4)$$

of $\mathcal{E}_2|_U$. Let e_1',e_2',e_3',e_4' be a different basis of $\mathbb{F}_{32003}^4$ such that e_3 and e_4 span the same 2-plane as e_3' and e_4'. We define $\tilde{e}_i'$ and $\tilde{\phi}_i'$ for $i=1,2,3,4$ as before.

We want to show that the determinants $\det_1$ and $\det_2$ of the two base change matrices relating $\tilde{\phi}_3\otimes\tilde{e}_1,\tilde{\phi}_4\otimes\tilde{e}_1,\tilde{\phi}_3\otimes\tilde{e}_2,\tilde{\phi}_4\otimes\tilde{e}_2$ to $\tilde{\phi}_3'\otimes\tilde{e}_1',\tilde{\phi}_4'\otimes\tilde{e}_1',\tilde{\phi}_3'\otimes\tilde{e}_2',\tilde{\phi}_4'\otimes\tilde{e}_2'$ and $(\tilde{\phi}_3^2,0),(\tilde{\phi}_3\tilde{\phi}_4,0),(\tilde{\phi}_4^2,0),(0,\tilde{\phi}_3\wedge\tilde{\phi}_4)$ to $(\tilde{\phi}_3'^2,0),(\tilde{\phi}_3'\tilde{\phi}_4',0),(\tilde{\phi}_4'^2,0),(0,\tilde{\phi}_3'\wedge\tilde{\phi}_4')$, respectively, are both squares, because then the determinant relating the bases of

$$T\,\mathrm{Gr}(2,4)^*|_U\otimes\mathcal{E}_2|_U\cong \mathrm{Hom}(T\,\mathrm{Gr}(2,4)|_U,\mathcal{E}_2|_U)$$

is a square.

By the proof [8, Lemma 42] $\det_1$ is a square. As in [8, Lemma 42] we write $\tilde{e}_3'=a\tilde{e}_3+b\tilde{e}_4$ and $\tilde{e}_4'=c\tilde{e}_3+d\tilde{e}_4$. Then

$$\tilde{\phi}_3'=A\tilde{\phi}_3+C\tilde{\phi}_4+\text{an element in the span of }\tilde{\phi}_1,\tilde{\phi}_2$$

and

$$\tilde{\phi}_4'=B\tilde{\phi}_3+D\tilde{\phi}_4+\text{an element in the span of }\tilde{\phi}_1,\tilde{\phi}_2$$

where $\begin{pmatrix}A&B\\C&D\end{pmatrix}$ is the inverse of $\begin{pmatrix}a&b\\c&d\end{pmatrix}$. The determinant relating $\tilde{\phi}_3^2,\tilde{\phi}_3\tilde{\phi}_4,\tilde{\phi}_4^2$ to $\tilde{\phi}_3'^2,\tilde{\phi}_3'\tilde{\phi}_4',\tilde{\phi}_4'^2$ is $(AD-BC)^3$ and the determinant relating $\tilde{\phi}_3\wedge\tilde{\phi}_4$ to $\tilde{\phi}_3'\wedge\tilde{\phi}_4'$ is $AD-BC$. Their product $\det_2=(AD-BC)^4$ is a square. □

We compute $e^{\mathbb{A}^1}(\mathcal{E}_3)$.

```
i35 : F2 = random(2,R);
i36 : a5 = random(1,R);
i37 : b5 = random(1,R);
i38 : s5 = lift((last coefficients m a5)_(0,0)*
(last coefficients m b5)_(1,0)
-(last coefficients m a5)_(1,0)*
(last coefficients m b5)_(0,0),C);
i39 : Q = sub(ideal flatten entries last coefficients m F2, C);
i40 : I3 = Q+ideal(s5);
i41 : J3 = determinant jacobian I3;
i42 : traceForm(C,I3,J3,FF)
```

It is a rank 4 form of discriminant $1\in\mathbb{F}_{32003}^*/(\mathbb{F}_{32003}^*)^2$. When we compute the form over the real numbers $\mathbb{R}$ (this can be done similarly as in subsubsection 2.3.3), we get a form of signature 0. Hence, we can use [1, Theorem 5.8] to conclude that $e^{\mathbb{A}^1}(\mathcal{E}_3)=2\mathbb{H}$.

Remark 7 Let E be a vector bundle that splits up as a direct sum of vector bundles, i.e. $E=E'\oplus E''$. It follows from [17, Proposition 12] that the $\mathbb{A}^1$-Euler number of E is a multiple of $\mathbb{H}=\langle 1\rangle+\langle -1\rangle$ if the rank of E' or E'' is odd. Hence, it is no surprise that we get a multiple of $\mathbb{H}$ in the calculation above.

2.3.3 Signature of $e^{\mathbb{A}^1}(\mathcal{E}) = e^{\mathbb{A}^1}(\mathrm{Sym}^3\,\mathcal{S}^*_{\mathbb{Q}})$

Let G be a degree 3 homogeneous polynomial in 4 variables with coefficients in $\mathbb{Q}$. For a general G the corresponding section σ_G of $\mathcal{E}$ will have finitely many zeros. We use the random function in Macaulay2 to generate a general degree 3 homogeneous polynomial.

```
i43 : R2 = QQ[Y0,Y1,Y2,Y3];
i44 : G = random(3,R2);
```

We compute $e^{\mathbb{A}^1}(\mathcal{E}_{\mathbb{Q}}, \sigma_G) \in \mathrm{GW}(\mathbb{Q})$. Base change yields a form over $\mathbb{R}$ of which we compute the signature as the number of positive eigenvalues minus the negative eigenvalues. Exactly as before, we restrict $\sigma_G : \mathrm{Gr}(2,4) \to \mathrm{Sym}^3\,\mathcal{S}^*$ to
$\mathrm{Spec}\, C_2 := \mathrm{Spec}(\mathbb{Q}[y_1,y_2,y_3,y_4]) \subset \mathrm{Gr}(2,4)$ and get $(g_1,g_2,g_3,g_4) : \mathbb{A}^4_{\mathbb{Q}} \to \mathbb{A}^4_{\mathbb{Q}}$ and let $I_4 = (g_1,g_2,g_3,g_4)$.

```
i45 : C2 = QQ[y1,y2,y3,y4];
i46 : S2 = C2[r];
i47 : g2 = {y1*r+y2,y3*r+y4,r,1};
i48 : m2 = map(S2,R2,g2);
i49 : I4 = sub(ideal flatten entries last coefficients m2 G, C2);
i50 : J4 = determinant jacobian I4;
```

We compute the trace form $\mathrm{Tr}_{(C_2/I_4)/\mathbb{Q}}(\langle (J_4)_{C_2/I_4}\rangle)$ where J_4 is the jacobian element of I_4 which is a 27×27-matrix with values in $\mathbb{Q}$. Viewing it as a form over $\mathbb{R}$, its signature is equal to the number of positive eigenvalues minus the number of negative eigenvalues because any real symmetric matrix can be diagonalized orthogonally.

```
i51 : Sol = traceForm(C2,I4,J4,QQ);
i52 : E = eigenvalues Sol;
i53 : sgn=0;
i54 : i=0;
i55 : while i<rk do(if E_i<0 then sgn=sgn-1 else sgn=sgn+1; i=i+1)
```

The signature is 3.

```
i56 : sgn

o56 = 3
```

So we know that the signature of $e^{\mathbb{A}^1}(\mathcal{E})$ is $n_{\mathbb{R}} = 3$ and its rank $n_{\mathbb{C}} = 27$. Since the discriminant of $e^{\mathbb{A}^1}(\mathcal{E}_{\mathbb{F}_{32003}}) \in \mathrm{GW}(\mathbb{F}_{32003})$ is a square (and 2 is not a square in $\mathbb{F}_{32003}$), we can conclude that $e^{\mathbb{A}^1}(\mathcal{E}) = e^{\mathbb{A}^1}(\mathrm{Sym}^3\,\mathcal{S}^*)$ is of form (2) and not (3), that is

$$e^{\mathbb{A}^1}(\mathcal{E}) = 15\langle 1\rangle + 12\langle -1\rangle.$$

2.3.4 Singular elements on a pencil of degree d hypersurfaces in $\mathbb{P}^3$

Let $\{F_t = t_0F_0 + t_1F_1 = 0\} \subset \mathbb{P}^3 \times \mathbb{P}^1$ be a pencil of degree d surfaces in $\mathbb{P}^3$. A surface in the pencil is singular if there is a point on the surface on which all 4 partial derivatives vanish simultaneously. Consider the vector bundle $\mathcal{F} := \bigoplus_{i=1}^4 \pi_1^*(\mathcal{O}_{\mathbb{P}^3}(d-1)) \otimes \pi_2^*(\mathcal{O}_{\mathbb{P}^1}(1)) \to \mathbb{P}^3 \times \mathbb{P}^1$ where $\pi_1 : \mathbb{P}^3 \times \mathbb{P}^1 \to \mathbb{P}^3$ and $\pi_2 : \mathbb{P}^3 \times \mathbb{P}^1 \to \mathbb{P}^1$ are the projections to the first and second factor, respectively. A pencil $X_t = \{F_t = t_0F_0 + t_1F_1 = 0\} \subset \mathbb{P}^3 \times \mathbb{P}^1$ defines a section $\sigma = (\frac{\partial F_t}{\partial X_0}, \dots, \frac{\partial F_t}{\partial X_3})$ of this bundle where $X_0, \dots, X_3$ are the coordinates on $\mathbb{P}^3$. A general singular hypersurface of degree d has a unique singularity which is an ordinary double point by [5, Proposition 7.1 (b)] and, whence, the zeros of σ are simple and count the singular elements on the pencil X_t. The bundle $\mathcal{F}$ is relatively orientable since $\mathcal{F}$ and $\mathbb{P}^3 \times \mathbb{P}^1$ are orientable, and we can enrich the count of singular elements on the pencil over $\mathrm{GW}(k)$.

Let $\mathbb{A}^3 \cong U_0 \subset \mathbb{P}^3$ and $\mathbb{A}^1 \cong V_0 \subset \mathbb{P}^1$ be the open affine subsets where X_0 and t_0 not vanish and let $\mathbb{A}^4 \cong U := U_0 \times V_0 \subset \mathbb{P}^3 \times \mathbb{P}^1$. One can show that U is compatible with the relative orientation of $\mathcal{F}$ in the same manner as in [13, Lemma 3.10].

Example 1 We provide the code for $d = 2$ over the field $\mathbb{F}_{32003}$.

```
i57 : F0 = random(2,R);
i58 : F1 = random(2,R);
i59 : T = R[t];
i60 : Ft = F0+t*F1;
i61 : D0 = diff(X0, Ft);
i62 : D1 = diff(X1, Ft);
i63 : D2 = diff(X2, Ft);
i64 : D3 = diff(X3, Ft);
i65 : C3 = FF[x1,x2,x3,t];
i66 : m3 = map(C3,T,{t,1,x1,x2,x3});
i67 : I5 = ideal(m3 D0,m3 D1,m3 D2,m3 D3);
i68 : J5 = determinant jacobian I5;
i69 : traceForm(C3,I5,J5,FF)
```

For the enriched count of singular elements on a pencil of degree 2 surfaces in $\mathbb{P}^3$ we get a form of rank 4, discriminant $1 \in \mathbb{F}_{32003}^*/(\mathbb{F}_{32003}^*)^2$ and signature 0, that is the form $2\mathbb{H}$. For $d = 3$, i.e., the enriched count of singular elements on a pencil of cubic surfaces, we get $16\mathbb{H}$ and for $d = 4$, $54\mathbb{H}$.

Remark 8 Again we know by [17, Proposition 12] that we get a multiple of the hyperbolic form $\mathbb{H} = \langle 1\rangle + \langle -1\rangle$.

Remark 9 Proposition 7.4 in [5] computes the number of singular elements on a pencil of degree d hypersurfaces in $\mathbb{P}^n$ to be $(n+1)(d-1)^n$. Whenever n is odd this count can be enriched in $\mathrm{GW}(k)$ to the form $\frac{(n+1)(d-1)^n}{2}\mathbb{H}$ by [17, Proposition 12]. One checks that this coincides with our count for $n = 3$ and $d = 2, 3, 4$.

Remark 10 Levine finds a formula [11, Corollary 10.4] that counts singular elements in a family as the sum the of $\mathbb{A}^1$-Milnor numbers of the singularities (see subsection 3.2 for the definition of $\mathbb{A}^1$-Minor numbers). It would be interesting to find a geometric interpretation for the local indices in our count and compare our result to Levine's count.

3 EKL-class

EKL is short for *Eisenbud-Khimshiashvili-Levine* who computed the local degree of non-simple, isolated zeros as the signature of a certain non-degenerate symmetric bilinear form (a representative of the EKL-class) over $\mathbb{R}$ in [6] and [10]. Eisenbud asked whether the class represented by the EKL-form which is defined in purely algebraic terms, had a meaningful interpretation over an arbitrary field k. His question was answered affirmatively in [9] where it is shown that the EKL-class is equal to the local $\mathbb{A}^1$-degree.

We recall the definition of the *EKL-class* from [9]. Let k be a field. Assume that $f = (f_1, \dots, f_n) : \mathbb{A}^n_k \to \mathbb{A}^n_k$ has an isolated zero at the origin and let $\mathcal{Q} := \frac{k[x_1,\dots,x_n]_{(x_1,\dots,x_n)}}{(f_1,\dots,f_n)}$. Define $E := \det a_{ij}$ where the $a_{ij} \in k[x_1, \dots, x_n]$ are chosen such that

$$f_i = f_i(0) + \sum_{i=1}^{n} a_{ij}x_j \overset{f_i(0)=0}{=} \sum_{i=1}^{n} a_{ij}x_j.$$

We call E the *distinguished socle element* since it generates the socle of $\mathcal{Q}$ (that is the sum of the minimal nonzero ideals) when f has an isolated zero at the origin [9, Lemma 4].

Remark 11 Let $J = \det \frac{\partial f_i}{\partial x_j}$ be the jacobian element. By [15, Korollar 4.7] $J = \mathrm{rank}_k\, \mathcal{Q} \cdot E$.

Let $\phi : \mathcal{Q} \to k$ be a k-linear functional which sends E to 1.

Definition 4 The *EKL-class* of f is the class of $\beta_\phi : \mathcal{Q} \times \mathcal{Q} \to k$ defined by $\beta_\phi(a, b) = \phi(ab)$ in $\mathrm{GW}(k)$.

Remark 12 By [9, Lemma 6] the EKL-class is well-defined, i.e., it does not depend on the choice of ϕ and β_ϕ is non-degenerate. One can for example choose a k-basis $b_1, \dots, b_{n-1}, E$ for $\mathcal{Q}$ and choose $\phi(b_i) = 0$ and $\phi(E) = 1$.

Table 1 Du Val singularities

Singularity	Equation f	$\mu^{\mathbb{A}^1}(f)$ = EKL-class of grad(f) $\in$ GW($\mathbb{Q}$)
A_n, n odd	$x^2+y^2+z^{n+1}$	$\frac{n-1}{2}\mathbb{H}+\langle n+1\rangle$
A_n, n even	$x^2+y^2+z^{n+1}$	$\frac{n}{2}\mathbb{H}$
D_n, $n>1$ odd	$x^2+y^2z+z^{n-1}$	$\frac{n-1}{2}\mathbb{H}+\langle -1\rangle$
D_n, n even	$x^2+y^2z+z^{n-1}$	$\frac{n-2}{2}\mathbb{H}+\langle -1\rangle+\langle n-1\rangle$
E_6	$x^2+y^3+z^4$	$3\mathbb{H}$
E_7	$x^2+y^3+yz^3$	$3\mathbb{H}+\langle -6\rangle$
E_8	$x^2+y^3+z^5$	$4\mathbb{H}$

3.1 EKL-code

The following code computes the EKL-form of $f:\mathbb{A}^n\to\mathbb{A}^n$ with one isolated zero at the origin when the characteristic of k does not divide $\mathrm{rank}_k\, \mathcal{Q}$. The input is a triple (C, I, FF) where the ideal $I=(f_1,\ldots,f_n)\subset C=FF[x_1,\ldots,x_n]$ which is a complete intersection and the output is the EKL-form.

```
i70: EKL=(C,I,FF)->(r=degree I;
B=basis(C/I);
B2=mutableMatrix B;
J=determinant jacobian I;
toVector = q -> last coefficients(q, Monomials=>B);
E=J_(C/I)/r;
p=0;j=0; while j<r do (if (toVector E)_(j,0)!=0 then p=j;j=j+1);
B2_(0,p):=E;
B2=matrix(B2);
Q=transpose B2 * B2;
T=mutableIdentity(C/I,r);
i=0;while i<r do (T_(i,p)=(toVector E)_(i,0) ; i=i+1);
T=matrix T;
T1=T^(-1);
linear = v -> v_(p,0);
M=matrix applyTable(entries Q,q->lift(linear(T1*(toVector q)),FF));
M)
```

3.2 $\mathbb{A}^1$-Milnor numbers

Kass and Wickelgren define and compute several $\mathbb{A}^1$-*Milnor numbers* as an application of the EKL-form in [9]. Let $0\in X=\{f=0\}\subset\mathbb{A}^n$ be a hypersurface with an isolated singularity at the origin. Then the $\mathbb{A}^1$-Milnor number of X is

$$\mu^{\mathbb{A}^1}(f):=\deg_0^{\mathbb{A}^1}(\mathrm{grad}(f)).$$

Kass and Wickelgren show that the $\mathbb{A}^1$-Milnor number is an invariant of the singularity. When n is even $\mu^{\mathbb{A}^1}(f)$ counts the nodes to which X bifurcates (see [9] for more details). They compute the $\mathbb{A}^1$-Milnor numbers of *ADE singularities*.

3.2.1 Du Val singularities

We compute the EKL class of Du Val singularities, that is simple singularities in 3 variables, in Table 1.

Example 2 As an example we give the computation for E_6.

```
i71 : C4 = QQ[x,y,z];
i72 : f = x^2+y^3+z^3*y;
i73 : I6 = ideal(diff(x,f),diff(y,f),diff(z,f));
```

We get the following EKL-form.

```
i74 : EKL(C4,I6,QQ)
```

```
o74 = | 0 0    0    1 0    0    0    |
      | 0 0    0    0 1/18 0    0    |
      | 0 0    0    0 0    1/18 0    |
      | 1 0    0    0 0    0    0    |
      | 0 1/18 0    0 0    0    0    |
      | 0 0    1/18 0 0    0    0    |
      | 0 0    0    0 0    0    -1/6 |

                 7        7
o74 : Matrix QQ  <--- QQ
```

It is easy to see that this is $3\mathbb{H} + \langle -6 \rangle$.

Acknowledgements
I would like to thank Kirsten Wickelgren for introducing me to the topic and her excellent guidance and feedback on this project. I am also very grateful to Anton Leykin for introducing me to Macaulay2. I gratefully acknowledge support by the RCN Frontier ResearchGroup Project no. 250399 "Motivic Hopf Equations."

Funding
Open Access funding enabled and organized by Projekt DEAL.

A More $\mathbb{A}^1$-Milnor numbers

We provide $\mathbb{A}^1$-Milnor numbers of some Fuchsian singularities (see [4]) in Table 2.

Table 2 Fuchsian singularities

Singularity	Equation f	$\mu^{\mathbb{A}^1}(f)$ = EKL-class of grad$(f) \in \mathrm{GW}(\mathbb{Q})$
E_{12}	$x^7 + y^3 + z^2$	$6\mathbb{H}$
Z_{11}	$x^5 + xy^3 + z^2$	$5\mathbb{H} + \langle -6 \rangle$
Q_{10}	$x^4 + y^3 + xz^2$	$5\mathbb{H}$
E_{13}	$x^5y + y^3 + z^2$	$6\mathbb{H} + \langle -10 \rangle$
Z_{12}	$x^4y + xy^3 + z^2$	$5\mathbb{H} + \langle -22 \rangle + \langle -66 \rangle$
Q_{11}	$x^3y + y^3 + xz^2$	$5\mathbb{H} + \langle 2 \rangle$
W_{12}	$x^5 + y^4 + z^2$	$6\mathbb{H}$
S_{11}	$x^4 + y^2z + xz^2$	$5\mathbb{H} + \langle -2 \rangle$
E_{14}	$x^8 + y^3 + z^2$	$7\mathbb{H}$
Z_{13}	$x^6 + xy^3 + z^2$	$6\mathbb{H} + \langle -6 \rangle$
Q_{12}	$x^5 + y^3 + xz^2$	$6\mathbb{H}$
W_{13}	$x^4y + y^4 + z^2$	$6\mathbb{H} + \langle -2 \rangle$
S_{12}	$x^3y + y^2z + xz^2$	$6\mathbb{H}$
U_{12}	$x^4 + y^3 + z^3$	$6\mathbb{H}$
$J_{0,3}$	$x^9 + y^3 + z^2$	$8\mathbb{H}$
$Z_{1,0}$	$x^7 + xy^3 + z^2$	$7\mathbb{H} + \langle -6 \rangle$
$Q_{2,0}$	$x^6 + y^3 + xz^2$	$7\mathbb{H}$
$W_{1,0}$	$x^6 + y^4 + z^2$	$7\mathbb{H} + \langle 3 \rangle$
$S_{1,0}$	$x^5 + zy^2 + xz^2$	$7\mathbb{H}$
$U_{1,0}$	$x^3y + y^3 + z^3$	$7\mathbb{H}$
W_{12}	$x^5 + y^4 + z^2$	$6\mathbb{H}$
$NA^1_{0,0}$	$x^5 + y^5 + z^2$	$8\mathbb{H}$
$VNA^1_{0,0}$	$x^4 + y^4 + yz^2$	$7\mathbb{H} + \langle -2 \rangle$
$J_{4,0}$	$x^{12} + y^3 + z^2$	$11\mathbb{H}$
$Z_{2,0}$	$x^{10} + xy^3 + z^2$	$10\mathbb{H} + \langle -6 \rangle$
$Q_{3,0}$	$x^9 + y^3 + xz^2$	$10\mathbb{H}$
$X_{2,0}$	$x^8 + y^4 + z^2$	$10\mathbb{H} + \langle 1 \rangle$
$S^*_{2,0}$	$x^7 + y^2z + xz^2$	$10\mathbb{H}$

Table 2 continued

$U^*_{2,0}$	$x^6 + y^3 + z^3$	$10\mathbb{H}$
	$x^6 + y^6 + z^2$	$12\mathbb{H} + \langle 2\rangle$
	$x^5 + y^5 + xz^2$	$12\mathbb{H}$
	$x^4 + y^4 + z^4$	$10\mathbb{H} + \langle 1\rangle$

B An example of lines on a cubic

As an example, we provide the Gram matrix of $e^{\mathbb{A}^1}(\mathcal{E}_{\mathbb{F}_{11}}, \sigma_H)$ for

$$H = Z_0^3 - Z0^2Z_1 - Z_1^2Z_2 + Z_0Z_2^2 - 2Z_1Z_2^2 - 2Z_0^2Z_3 - Z_0Z_1Z_3 \\ -X_1^2X_3 + X_1X_2X_3 + X_1X_3^2 + 2X_2X_3^2,$$

that is, the count of lines on the cubic surface $\{H = 0\} \subset \mathbb{P}^3_{\mathbb{F}_{11}}$.

```
i75: P2 = 11;
i76 : FF2 = ZZ/P2;
i77 : R3 = FF2[Z0,Z1,Z2,Z3];
i78 : H = Z0^3-Z0^2*Z1-Z1^2*Z2+Z0*Z2^2-2*Z1*Z2^2-
    2*Z0^2*Z3-Z0*Z1*Z3-Z1^2*Z3+Z1*Z2*Z3+Z1*Z3^2+2*Z2*Z3^2;
i79 : C5 = FF2[z1,z2,z3,z4];
i80 : S3 = C5[u];
i81 : g3 = {z1*u+z2,z3*u+z4,u,1};
i82 : m4 = map(S3,R3,g3);
i83 : I7 = sub(ideal flatten entries last coefficients m4 H, C5);
i84 : L2=minimalPrimes I7;
i85 : n2=length L2;
i86 : J7=determinant jacobian I7;
```

There are 5 lines on X.

```
i87 : n2=length L2

o87 = 5
```

Let $L_1, \ldots, L_5$ be the fields of definitions of the 5 lines. We compute the trace forms of $\mathrm{Tr}_{L_j/\mathbb{F}_{11}}\langle J_{L_j}\rangle$ for $j = 1, \ldots, 5$ and sum them up to get $e^{\mathbb{A}^1}(\mathcal{E}_{\mathbb{F}_{11}}, \sigma_F) \in \mathrm{GW}(\mathbb{F}_{11})$.

```
i88 : Sol2 = traceForm(C5,L2_0,J7,FF2);
i89 : j=1;
i90 : while j<n2 do (Sol2=Sol2++traceForm(C5,L2_j,J7,FF2);j=j+1);
i91 = Sol2

o91 = | -3 0 0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  |
      | 0  1 2  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  |
      | 0  2 -3 0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  |
      | 0  0 0  -3 -4 2  0  0  0  2  -2 0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  |
      | 0  0 0  -4 5  -5 -3 3  -5 2  3  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  |
      | 0  0 0  2  -5 -5 -4 -4 -5 0  5  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  |
      | 0  0 0  0  -3 -4 5  -5 -5 5  -2 0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  |
      | 0  0 0  0  3  -4 -5 -5 4  0  1  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  |
      | 0  0 0  0  -5 -5 -5 4  3  1  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  |
      | 0  0 0  2  2  0  5  0  1  -2 2  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  |
      | 0  0 0  -2 3  5  -2 1  0  2  -5 0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  |
      | 0  0 0  0  0  0  0  0  0  0  0  -3 -3 2  -4 4  -1 4  0  0  0  0  0  0  0  0  |
      | 0  0 0  0  0  0  0  0  0  0  0  -3 2  -5 -1 2  0  0  4  0  0  0  0  0  0  0  |
      | 0  0 0  0  0  0  0  0  0  0  0  2  -5 4  -3 1  5  -4 1  0  0  0  0  0  0  0  |
      | 0  0 0  0  0  0  0  0  0  0  0  -4 -1 -3 -2 5  5  1  1  0  0  0  0  0  0  0  |
      | 0  0 0  0  0  0  0  0  0  0  0  4  2  1  5  -4 -1 -1 -2 0  0  0  0  0  0  0  |
      | 0  0 0  0  0  0  0  0  0  0  0  -1 0  5  5  -1 -1 -2 -2 0  0  0  0  0  0  0  |
      | 0  0 0  0  0  0  0  0  0  0  0  4  0  -4 1  -1 -2 0  2  0  0  0  0  0  0  0  |
      | 0  0 0  0  0  0  0  0  0  0  0  0  4  1  1  -2 -2 2  -5 0  0  0  0  0  0  0  |
      | 0  0 0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  -3 -3 0  -2 -4 4  3  |
```

```
        | 0  0 0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  -3 4  -4 2  2  -4 -3 2  |
        | 0  0 0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  -3 -4 3  3  -4 -3 2  -5 |
        | 0  0 0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  2  3  4  -1 -5 -2 -3 |
        | 0  0 0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  -2 2  -4 -1 -4 -2 -4 -4 |
        | 0  0 0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  -4 -4 -3 -5 -2 3  -4 1  |
        | 0  0 0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  4  -3 2  -2 -4 -4 3  -2 |
        | 0  0 0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  0  3  2  -5 -3 -4 1  -2 -2 |

                27        27
o90 : Matrix FF   <--- FF
```

The sizes of the blocks are the degrees $[L_j : \mathbb{F}_{11}]$ of the field extension $L_j/\mathbb{F}_{11}$ for $j = 1, \ldots, 5$. So there is one rational line on X, one defined over a field extension of degree 2 and 3 lines defined over a field extension of degree 8 on X.

Received: 29 February 2020 Accepted: 14 May 2021 Published online: 10 June 2023

References

1. Bachmann, T., Wickelgren, K.: A1-Euler classes: six functors formalisms, dualities, integrality and linear subspaces of complete intersections (2020)
2. Brazelton, T., Burklund, R., McKean, S., Montoro, M., Opie, M.: The trace of the local $\mathbf{A}^1$-degree (2019)
3. Cayley, A.: On the Triple Tangent Planes of Surfaces of the Third Order, Cambridge Library Collection—Mathematics, vol. 1, pp. 445–456. Cambridge University Press, Cambridge (2009). https://doi.org/10.1017/CBO9780511703676.077
4. Ebeling, W.: The Poincaré series of some special quasihomogeneous surface singularities. Publ. Res. Inst. Math. Sci. **39**(2), 393–413 (2003)
5. Eisenbud, D., Harris, J.: 3264 and All That—A Second Course in Algebraic Geometry. Cambridge University Press, Cambridge (2016). https://doi.org/10.1017/CBO9781139062046
6. Eisenbud, D., Levine, H.I.: An algebraic formula for the degree of a C^∞ map germ. Ann. Math. (2) **106**(1), 19–44 (1977). https://doi.org/10.2307/1971156
7. Finashin, S., Kharlamov, V.: Abundance of 3-planes on real projective hypersurfaces. Arnold Math. J. **1**(2), 171–199 (2015). https://doi.org/10.1007/s40598-015-0015-5
8. Kass, J.L., Wickelgren, K.: An Arithmetic Count of the Lines on a Smooth Cubic Surface. arXiv e-prints arXiv:1708.01175 (2017)
9. Kass, J.L., Wickelgren, K.: The class of Eisenbud–Khimshiashvili–Levine is the local $\mathbf{A}^1$-Brouwer degree. Duke Math. J. **168**(3), 429–469 (2019). https://doi.org/10.1215/00127094-2018-0046
10. Khimshiashvili, G.N.: The local degree of a smooth mapping. Sakharth. SSR Mecn. Akad. Moambe **85**(2)
11. Levine, M.: Toward an enumerative geometry with quadratic forms (2017)
12. Levine, M.: Motivic Euler characteristics and Witt-valued characteristic classes. Nagoya Math. J. (2019). https://doi.org/10.1017/nmj.2019.6
13. McKean, S.: An enrichment of Bézout's theorem (2020)
14. Okonek, C., Teleman, A.: Intrinsic signs and lower bounds in real algebraic geometry. J. Reine Angew. Math. **688**, 219–241 (2014). https://doi.org/10.1515/crelle-2012-0055
15. Scheja, G., Storch, U.: Uber Spurfunktionen bei vollst andigen Durchschnitten. J. Reine Angew. Math. **278**(279), 174–190 (1975)
16. Segre, B.: The Non-singular Cubic Surfaces. Oxford University Press, Oxford (1942)
17. Srinivasan, P., Wickelgren, K.: An arithmetic count of the lines meeting four lines in p3 (2018)

Publisher's Note